

General Relativity using Computer Algebra

Problem: for the spacetime metric,

$$g_{\mu, \nu} = \begin{bmatrix} -e^{\lambda(r)} & 0 & 0 & 0 \\ 0 & -r^2 & 0 & 0 \\ 0 & 0 & -r^2 \sin(\theta)^2 & 0 \\ 0 & 0 & 0 & e^{\nu(r)} \end{bmatrix}$$

a) Compute the trace of

$$Z_{\alpha}^{\beta} = \Phi R_{\alpha}^{\beta} + \mathcal{D}_{\alpha} \mathcal{D}^{\beta} \Phi + T_{\alpha}^{\beta}$$

where $\Phi \equiv \Phi(r)$ is some function of the radial coordinate, R_{α}^{β} is the Ricci tensor, \mathcal{D}_{α} is the covariant derivative operator and T_{α}^{β} is the stress-energy tensor

$$T_{\alpha, \beta} = \begin{bmatrix} 8 e^{\lambda(r)} \pi & 0 & 0 & 0 \\ 0 & 8 r^2 \pi & 0 & 0 \\ 0 & 0 & 8 r^2 \sin(\theta)^2 \pi & 0 \\ 0 & 0 & 0 & 8 e^{\nu(r)} \pi \varepsilon \end{bmatrix}$$

b) Compute the components of $W_{\alpha}^{\beta} \equiv$ the traceless part of Z_{α}^{β} of item **a)**

c) Compute an exact solution to the nonlinear system of differential equations conformed by the components of W_{α}^{β} obtained in **b)**

Background: The equations of items **a)** and **b)** appear in a paper from February/2013, "[Withholding Potentials, Absence of Ghosts and Relationship between Minimal Dilatonic Gravity and f\(R\) Theories](#)", by Plamen Fiziev, a Maple user. These equations model a problem in the context of a Brans-Dicke theory with vanishing parameter ω . The Brans-Dicke theory is in many respects similar to Einstein's theory, but the gravitational "constant" is not actually presumed to be constant - it can vary from place to place and with time - and the gravitational interaction is mediated by a scalar field. Both Brans-Dicke's and Einstein's theory of general relativity are generally held to be in agreement with observation.

The computations below aim at illustrating how this type of computation can be performed using computer algebra, and so they focus only on the algebraic aspects, not the physical interpretation of the results.

▼ **a) The trace of** $Z_{\alpha}^{\beta} = \Phi R_{\alpha}^{\beta} + \mathcal{D}_{\alpha} \mathcal{D}^{\beta} \Phi + T_{\alpha}^{\beta}$

The computations that follow make use of the latest version of the Physics package, available for download on the [Maple Physics: Research & Development](#) webpage.

> restart, with (Physics) : Physics:-Version() [2]
 2014, February 20, 15:8 hours (1.1)

Set the coordinates

> Setup(mathematicalnotation = true, coordinates = spherical)
 * Partial match of 'coordinates' against keyword 'coordinatesystems'
 Default differentiation variables for d_, D_ and dAlembertian are: {X = (r, θ, φ, t)}
 Systems of spacetime Coordinates are: {X = (r, θ, φ, t)}
 [coordinatesystems = {X}, mathematicalnotation = true] (1.2)

Introduce the line element and with it set the metric

> ds2 := exp(nu(r))dt² - exp(lambda(r))dr² - r² dtheta² - r² sin(theta)² dphi²
 $ds_2 := e^{v(r)} dt^2 - e^{\lambda(r)} dr^2 - r^2 d\theta^2 - r^2 \sin(\theta)^2 d\phi^2$ (1.3)

On the way, keep the display of functions compact and derivatives displayed indexed

> PDEtools:-declare(ds2)
 $\lambda(r)$ will now be displayed as λ
 $v(r)$ will now be displayed as v (1.4)

> Setup(metric = ds2)
 $[metric = \{(1, 1) = -e^\lambda, (2, 2) = -r^2, (3, 3) = -r^2 \sin(\theta)^2, (4, 4) = e^v\}]$ (1.5)

Verify we got the correct covariant metric components

> g_[]

$$g_{\mu, \nu} = \begin{bmatrix} -e^\lambda & 0 & 0 & 0 \\ 0 & -r^2 & 0 & 0 \\ 0 & 0 & -r^2 \sin(\theta)^2 & 0 \\ 0 & 0 & 0 & e^v \end{bmatrix}$$
 (1.6)

Define now the stress-energy tensor indicated with the problem. Among the different ways to introduce it (see [Physics:-Define](#)), we choose entering it as a matrix for the covariant components

> T[alpha, beta] = 8·Pi·Matrix(4, 4, [exp(lambda(r)), 0, 0, 0, 0, r², 0, 0, 0, 0, r²sin(theta)², 0, 0, 0, 0, epsilon exp(nu(r))])

$$T_{\alpha, \beta} = \begin{bmatrix} 8 \pi e^\lambda & 0 & 0 & 0 \\ 0 & 8 \pi r^2 & 0 & 0 \\ 0 & 0 & 8 \pi r^2 \sin(\theta)^2 & 0 \\ 0 & 0 & 0 & 8 \pi \epsilon e^v \end{bmatrix}$$
 (1.7)

> Define((1.7))
 Defined objects with tensor properties

$$\left\{ \mathcal{D}_\mu, \gamma_\mu, \sigma_\mu, R_{\mu, \nu}, R_{\mu, \nu, \alpha, \beta}, T_{\alpha, \beta}, C_{\mu, \nu, \alpha, \beta}, X_\mu, \partial_\mu, g_{\mu, \nu}, \Gamma_{\mu, \nu, \alpha}, G_{\mu, \nu}, \delta_{\mu, \nu}, \epsilon_{\alpha, \beta, \mu, \nu} \right\} \quad (1.8)$$

After this definition, for example, the component $T^{3,3}$ is given by

> $T[\sim 3, \sim 3]$

$$\frac{8 \pi}{r^2 \sin(\theta)^2} \quad (1.9)$$

and the trace of T can be computed via

> $T[\text{alpha}, \text{alpha}]$

$$T_\alpha^\alpha \quad (1.10)$$

> *SumOverRepeatedIndices*((1.10))

$$8 \pi \epsilon - 24 \pi \quad (1.11)$$

Having defined the metric and the Stress energy tensor, we can solve item **a)** in one go by defining a single tensor $Z_{\mu, \nu}$ as the tensorial equation

$$Z_\alpha^\beta = \Phi R_\alpha^\beta + \mathcal{D}_\alpha \mathcal{D}^\beta \Phi + T_\alpha^\beta$$

> *PDEtools:-declare*(Phi(r))

$$\Phi(r) \text{ will now be displayed as } \Phi \quad (1.12)$$

For illustration purposes we use the inert %D_ command, this also permits to see what we are entering before any computation is performed

> %Z[mu, nu] = Phi(r) Ricci[mu, nu] + %D_[mu](%D_[nu](Phi(r))) + T[mu, nu]

$$\%Z_{\mu, \nu} = \Phi R_{\mu, \nu} + \mathcal{D}_\mu \left(\mathcal{D}_\nu (\Phi) \right) + T_{\mu, \nu} \quad (1.13)$$

The covariant derivatives $\mathcal{D}_\mu \left(\mathcal{D}_\nu (\Phi(r)) \right)$ can be activated when desired, for example:

> *value*((1.13))

$$Z_{\mu, \nu} = \Phi R_{\mu, \nu} + g_\nu^1 \Phi_{,r} g_\mu^1 - \Gamma_{\mu, \nu}^\alpha \Phi_r g_\alpha^1 + T_{\mu, \nu} \quad (1.14)$$

(Note the occurrence of the Christofel symbols in the result above - more about them in the NOTES at the end of the section.)

We can now define $Z_{\mu, \nu}$ with the inert \mathcal{D} or with the evaluated form (1.14). To compare both approaches, we define both $Z_{\mu, \nu}$ and %Z_{μ,ν}

> *Define*((1.13), (1.14))

Defined objects with tensor properties

$$\left\{ \%Z_{\mu, \nu}, \mathcal{D}_\mu, \gamma_\mu, \sigma_\mu, R_{\mu, \nu}, R_{\mu, \nu, \alpha, \beta}, T_{\alpha, \beta}, C_{\mu, \nu, \alpha, \beta}, X_\mu, Z_{\mu, \nu}, \partial_\mu, g_{\mu, \nu}, \Gamma_{\mu, \nu, \alpha}, G_{\mu, \nu}, \delta_{\mu, \nu}, \epsilon_{\alpha, \beta, \mu, \nu} \right\} \quad (1.15)$$

The answer to **a)**, that is the trace of $Z_{\mu, \nu}$ is then

> $Z[\text{mu}, \text{mu}]$

$$Z_\mu^\mu \quad (1.16)$$

> *SumOverRepeatedIndices* ((1.16))

$$\begin{aligned} & \frac{\Phi \left(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r - 4 \lambda_r \right) e^{-\lambda}}{4 r} - e^{-\lambda} \Phi_{r,r} + \frac{\lambda_r e^{-\lambda} \Phi_r}{2} - 24 \pi \\ & - \frac{\Phi \left(\lambda_r e^{-\lambda} r - v_r e^{-\lambda} r - 2 e^{-\lambda} + 2 \right)}{r^2} - \frac{2 e^{-\lambda} \Phi_r}{r} \\ & + \frac{\Phi e^{-\lambda} \left(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r + 4 v_r \right)}{4 r} - \frac{v_r e^{-\lambda} \Phi_r}{2} + 8 \pi \epsilon \end{aligned} \quad (1.17)$$

For a closer look at the representation capabilities using inert functions, compare this result (1.17) with the equivalent one obtained with the differentiation operations inert

> %Z[mu, mu]

$$\%Z_{\mu}^{\mu} \quad (1.18)$$

> *SumOverRepeatedIndices* ((1.18))

$$\begin{aligned} & \frac{\Phi \left(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r - 4 \lambda_r \right) e^{-\lambda}}{4 r} + \partial_1(\partial^1(\Phi)) + \frac{\lambda_r \partial^1(\Phi)}{2} - 24 \pi \\ & - \frac{\Phi \left(\lambda_r e^{-\lambda} r - v_r e^{-\lambda} r - 2 e^{-\lambda} + 2 \right)}{r^2} + \partial_2(\partial^2(\Phi)) + \frac{2 \partial^1(\Phi)}{r} + \partial_3(\partial^3(\Phi)) \\ & + \frac{\cos(\theta) \partial^2(\Phi)}{\sin(\theta)} + \frac{\Phi e^{-\lambda} \left(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r + 4 v_r \right)}{4 r} + \partial_4(\partial^4(\Phi)) \\ & + \frac{v_r \partial^1(\Phi)}{2} + 8 \pi \epsilon \end{aligned} \quad (1.19)$$

Activate now the inert covariant ∂_{μ} and contravariant ∂^{μ} derivatives and compare with (1.17)

> *value*((1.19))

$$\begin{aligned} & \frac{\Phi \left(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r - 4 \lambda_r \right) e^{-\lambda}}{4 r} - e^{-\lambda} \Phi_{r,r} + \frac{\lambda_r e^{-\lambda} \Phi_r}{2} - 24 \pi \\ & - \frac{\Phi \left(\lambda_r e^{-\lambda} r - v_r e^{-\lambda} r - 2 e^{-\lambda} + 2 \right)}{r^2} - \frac{2 e^{-\lambda} \Phi_r}{r} \\ & + \frac{\Phi e^{-\lambda} \left(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r + 4 v_r \right)}{4 r} - \frac{v_r e^{-\lambda} \Phi_r}{2} + 8 \pi \epsilon \end{aligned} \quad (1.20)$$

> *normal*((1.17)-(1.20))

$$0 \quad (1.21)$$

NOTES

Part of the design of the Physics package is that as soon as you define the spacetime metric all the

general relativity tensors like the Ricci tensor R_{α}^{β} , or the Christoffel symbols $\Gamma_{\mu, \nu}^{\alpha}$ and the covariant derivative operator \mathcal{D}_{α} get automatically defined on background. For example, you do not need to additionally compute the Ricci tensor - instead, for instance the value of R_1^1 can be compute directly, via

> *Ricci*[1, ~1]

$$\frac{(-\lambda_r v_r r + v_r^2 r + 2 v_{r,r} r - 4 \lambda_r) e^{-\lambda}}{4 r} \quad (1.22)$$

All the covariant components at once:

> *Ricci*[]

$$R_{\mu, \nu} = \left[\left[\frac{\lambda_r v_r r - v_r^2 r - 2 v_{r,r} r + 4 \lambda_r}{4 r}, 0, 0, 0 \right], \right. \quad (1.23)$$

$$\left. \left[0, 1 + \frac{(-v_r r + r \lambda_r - 2) e^{-\lambda}}{2}, 0, 0 \right], \right.$$

$$\left. \left[0, 0, -\frac{(-2 + (v_r r - r \lambda_r + 2) e^{-\lambda}) \sin(\theta)^2}{2}, 0 \right], \right.$$

$$\left. \left[0, 0, 0, \frac{(2 v_{r,r} r + v_r (v_r r - r \lambda_r + 4)) e^{-\lambda + \nu}}{4 r} \right] \right]$$

The same happens with the Riemann, Einstein and Weyl tensors, as well as with the Christoffel symbols of the first and second kind, where the "kind" just depends on whether the indices are covariant or contravariant:

> *Christoffel*[alpha, mu, nu]

$$\Gamma_{\alpha, \mu, \nu} \quad (1.24)$$

> '*Christoffel*[2, 3, 3]' = *Christoffel*[2, 3, 3]

$$\Gamma_{2, 3, 3} = r^2 \sin(\theta) \cos(\theta) \quad (1.25)$$

> *g*[~beta, ~alpha] (1.24)

$$g^{\alpha, \beta} \Gamma_{\alpha, \mu, \nu} \quad (1.26)$$

> *Simplify*((1.26))

$$\Gamma_{\mu, \nu}^{\beta} \quad (1.27)$$

> '*Christoffel*[~2, 3, 3]' = *Christoffel*[~2, 3, 3]

$$\Gamma_{3, 3}^2 = -\sin(\theta) \cos(\theta) \quad (1.28)$$

All these tensors, or the ones you define using tensorial equations, admit the keywords nonzero and matrix to display the corresponding components, as well as the shortcut notation where the tensor is called with no indices returning all its covariant components:

> *Christoffel*[~2, alpha, beta, matrix]

$$\Gamma_{\alpha, \beta}^2 = \begin{bmatrix} 0 & \frac{1}{r} & 0 & 0 \\ \frac{1}{r} & 0 & 0 & 0 \\ 0 & 0 & -\sin(\theta) \cos(\theta) & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (1.29)$$

> 'Christoffel[2,~3, 3]' = Christoffel[2,~3, 3]

$$\Gamma_{2\ 3}^3 = -\frac{\cos(\theta)}{\sin(\theta)} \quad (1.30)$$

> 'Christoffel[2, 3,~3]' = Christoffel[2, 3,~3]

$$\Gamma_{2,3}^3 = -\frac{\cos(\theta)}{\sin(\theta)} \quad (1.31)$$

Also, when computing with any of these tensors (or the ones you define) without specifying values for the indices, the symmetries under permutation of the indices are automatically taken into account; for example for the Christoffel symbols, permuting the last two indices does not change their value, so

$$\text{Christoffel}[\alpha, \mu, \nu] - \text{Christoffel}[\alpha, \nu, \mu] = 0 \quad (1.32)$$

▼ b) The components of $W_{\alpha}^{\beta} \equiv$ the traceless part of Z_{α}^{β}

This problem can be solved in one step, by defining a single tensor $W_{\mu, \nu}$ with the the traceless part of $Z_{\mu, \nu}$. We use the standard definition of traceless

$$\text{> } W[\mu, \nu] = Z[\mu, \nu] - \frac{Z[\alpha, \alpha]}{4} g_{\mu, \nu}$$

$$W_{\mu, \nu} = Z_{\mu, \nu} - \frac{Z_{\alpha}^{\alpha} g_{\mu, \nu}}{4} \quad (2.1)$$

> Define((2.1))

Defined objects with tensor properties

$$\left\{ \%Z_{\mu, \nu}, \mathcal{D}_{\mu}, \gamma_{\mu}, \sigma_{\mu}, R_{\mu, \nu}, R_{\mu, \nu, \alpha, \beta}, T_{\alpha, \beta}, W_{\mu, \nu}, C_{\mu, \nu, \alpha, \beta}, X_{\mu}, Z_{\mu, \nu}, \partial_{\mu}, g_{\mu, \nu}, \Gamma_{\mu, \nu, \alpha}, G_{\mu, \nu}, \delta_{\mu, \nu}, \epsilon_{\alpha, \beta, \mu, \nu} \right\} \quad (2.2)$$

For example, for $W^{1,1}$

> W[~1, ~1]

$$\frac{\Phi \left(\lambda_r v_r r - v_r^2 r - 2 v_{r,r} r + 4 \lambda_r \right) e^{-2\lambda}}{4r} + (e^{-\lambda})^2 \Phi_{r,r} - \frac{e^{-2\lambda} \lambda_r \Phi_r}{2} + 8 \pi e^{-\lambda} \quad (2.3)$$

$$+ \frac{1}{4} \left(\left(-\frac{\Phi \left(\lambda_r v_r r - v_r^2 r - 2 v_{r,r} r + 4 \lambda_r \right) e^{-\lambda}}{4r} - e^{-\lambda} \Phi_{r,r} + \frac{\lambda_r e^{-\lambda} \Phi_r}{2} - 24 \pi \right) \right)$$

$$\begin{aligned}
& - \frac{\Phi \left(\lambda_r e^{-\lambda} r - v_r e^{-\lambda} r - 2 e^{-\lambda} + 2 \right)}{r^2} - \frac{2 e^{-\lambda} \Phi_r}{r} \\
& - \frac{\Phi e^{-\lambda} \left(\lambda_r v_r r - v_r^2 r - 2 v_{r,r} r - 4 v_r \right)}{4 r} - \frac{v_r e^{-\lambda} \Phi_r}{2} + 8 \pi \varepsilon \left. \vphantom{\frac{\Phi \left(\lambda_r e^{-\lambda} r - v_r e^{-\lambda} r - 2 e^{-\lambda} + 2 \right)}{r^2}} \right) e^{-\lambda}
\end{aligned}$$

Or the matrix components for the traceless mixed W_μ^ν

> $W[\text{mu}, \sim \text{nu}, \text{matrix}]$

$$\begin{aligned}
W_\mu^\nu = & \left[\left[\frac{1}{8 r^2} \left(\left(-6 \Phi_{r,r} r^2 + 2 \Phi v_{r,r} r^2 + \Phi v_r^2 r^2 - r \left(\Phi \lambda_r r - \Phi_r r + 4 \Phi \right) v_r \right. \right. \right. \quad (2.4) \\
& \left. \left. \left. + \left(3 r^2 \Phi_r - 4 r \Phi \right) \lambda_r + 4 \Phi_r r - 4 \Phi \right) e^{-\lambda} + 4 \Phi + \left(-16 \varepsilon - 16 \right) \pi r^2 \right), 0, 0, 0 \right], \right. \\
& \left[0, \frac{1}{8 r^2} \left(\left(2 \Phi_{r,r} r^2 - 2 \Phi v_{r,r} r^2 + \left(r^2 v_r - r^2 \lambda_r - 4 r \right) \Phi_r - \Phi \left(r^2 v_r^2 \right. \right. \right. \\
& \left. \left. \left. - r^2 \lambda_r v_r - 4 \right) \right) e^{-\lambda} - 4 \Phi + \left(-16 \varepsilon - 16 \right) \pi r^2 \right), 0, 0 \right], \\
& \left[0, 0, \frac{1}{8 r^2} \left(\left(2 \Phi_{r,r} r^2 - 2 \Phi v_{r,r} r^2 + \left(r^2 v_r - r^2 \lambda_r - 4 r \right) \Phi_r - \Phi \left(r^2 v_r^2 \right. \right. \right. \\
& \left. \left. \left. - r^2 \lambda_r v_r - 4 \right) \right) e^{-\lambda} - 4 \Phi + \left(-16 \varepsilon - 16 \right) \pi r^2 \right), 0 \right], \\
& \left. \left[0, 0, 0, \frac{1}{8 r^2} \left(\left(2 \Phi_{r,r} r^2 + 2 \Phi v_{r,r} r^2 + \Phi v_r^2 r^2 - r \left(\Phi \lambda_r r + 3 \Phi_r r \right. \right. \right. \right. \right. \\
& \left. \left. \left. - 4 \Phi \right) v_r + \left(-r^2 \Phi_r + 4 r \Phi \right) \lambda_r + 4 \Phi_r r - 4 \Phi \right) e^{-\lambda} + 4 \Phi + \left(48 \varepsilon + 48 \right) \pi r^2 \right) \right] \right]
\end{aligned}$$

Despite that $W[\text{mu}, \text{nu}]$ is traceless by construction (see (2.1)), one can verify that explicitly

> $W[\text{alpha}, \text{alpha}]$

$$W_\alpha^\alpha \quad (2.5)$$

> $\text{SumOverRepeatedIndices}((2.5))$

$$0 \quad (2.6)$$

▼ c) An exact solution for the nonlinear system of differential equations conformed by the components of W_α^β

Create an ODE system with the nonzero components of W_μ^ν

> $\text{ode}_{\text{sys}} := \text{map}(\text{rhs}, \text{rhs}(W[\text{mu}, \sim \text{nu}, \text{nonzero}]))$

$$\text{ode}_{\text{sys}} := \left\{ \frac{1}{8 r^2} \left(\left(2 \Phi_{r,r} r^2 - 2 \Phi v_{r,r} r^2 + \left(r^2 v_r - r^2 \lambda_r - 4 r \right) \Phi_r - \Phi \left(r^2 v_r^2 \right. \right. \right. \right. \quad (3.1)$$

$$\begin{aligned}
& -r^2 \lambda_r v_r - 4) e^{-\lambda} - 4 \Phi + (-16 \varepsilon - 16) \pi r^2), \frac{1}{8 r^2} \left((-6 \Phi_{r,r} r^2 \right. \\
& + 2 \Phi v_{r,r} r^2 + \Phi v_r^2 r^2 - r (\Phi \lambda_r r - \Phi_r r + 4 \Phi) v_r + (3 r^2 \Phi_r - 4 r \Phi) \lambda_r \\
& + 4 \Phi_r r - 4 \Phi) e^{-\lambda} + 4 \Phi + (-16 \varepsilon - 16) \pi r^2), \frac{1}{8 r^2} \left((2 \Phi_{r,r} r^2 \right. \\
& + 2 \Phi v_{r,r} r^2 + \Phi v_r^2 r^2 - r (\Phi \lambda_r r + 3 \Phi_r r - 4 \Phi) v_r + (-r^2 \Phi_r + 4 r \Phi) \lambda_r \\
& \left. + 4 \Phi_r r - 4 \Phi) e^{-\lambda} + 4 \Phi + (48 \varepsilon + 48) \pi r^2) \right\}
\end{aligned}$$

> Case := simplify([PDEtools:-casesplit(ode_sys)], size) :

There are three cases

> nops(Case)

3

(3.2)

> Case[1]

$$\left[e^{-\lambda} = \left(-16 r \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi_r + 8 \left(r^3 \pi (\varepsilon + 1) v_r + \frac{\Phi}{2} + (2 \varepsilon \right. \right. \right. \quad (3.3)$$

$$\left. + 2) \pi r^2 \right) \Phi \right) / \left(-r^2 \Phi (v_r r - 2) \Phi_{r,r} + (\Phi_r r - 2 \Phi) (\Phi v_{r,r} r^2 + (-r^2 v_r \right.$$

$$\left. + 2 r) \Phi_r + \Phi (r^2 v_r^2 - 2) \right)), \lambda_r = \left(-16 r \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi_{r,r} \right.$$

$$\left. + 8 r^3 \pi \Phi (\varepsilon + 1) v_{r,r} + 4 r^3 \pi \Phi (\varepsilon + 1) v_r^2 - 8 \left(-\frac{\Phi_r r}{8} + \frac{\Phi}{4} + (\varepsilon \right. \right.$$

$$\begin{aligned}
& + 1) r^2 \pi \left) \Phi v_r + 16 r \pi (\varepsilon + 1) (\Phi_r r - \Phi) \right) / \left(-8 r \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi_r \right. \\
& + 4 \left(r^3 \pi (\varepsilon + 1) v_r + \frac{\Phi}{2} + (2 \varepsilon + 2) \pi r^2 \right) \Phi \left. \right), v_{r,r,r} = \left(4 r^2 \left(-2 r \left(\frac{\Phi}{8} \right. \right. \right. \\
& + (\varepsilon + 1) r^2 \pi \left. \right) \Phi_r + \left(r^3 \pi (\varepsilon + 1) v_r + \frac{\Phi}{2} + (2 \varepsilon + 2) \pi r^2 \right) \Phi \left. \right) (v_r r \\
& - 2) \Phi \Phi_{r,r,r} + 24 r^3 (v_r r - 2) \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi \Phi_{r,r}^2 \\
& - 8 r \left(\frac{1}{2} \left(3 r^2 \left(2 r \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi_r + \left(r^3 \pi (\varepsilon + 1) v_r - \frac{\Phi}{2} + (-6 \varepsilon \right. \right. \right. \right. \\
& - 6) \pi r^2 \left. \right) \Phi \left. \right) v_{r,r} \left. \right) + \left(\frac{r^2}{4} - \frac{v_r r^3}{8} \right) \Phi_r^2 + r \left(r^2 \left(\frac{3 \Phi}{8} + (\varepsilon + 1) r^2 \pi \right) v_r^2 \right. \\
& + 6 r \left(\frac{\Phi}{24} + (\varepsilon + 1) r^2 \pi \right) v_r - \frac{5 \Phi}{4} + (-10 \varepsilon - 10) \pi r^2 \left. \right) \Phi_r + \left(r^5 \pi (\varepsilon \right. \\
& + 1) v_r^3 - 6 r^2 \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) v_r^2 - 3 r \left(-\frac{\Phi}{12} + (\varepsilon + 1) r^2 \pi \right) v_r + \Phi \\
& + (10 \varepsilon + 10) \pi r^2 \left. \right) \Phi \left. \right) \Phi \Phi_{r,r} + 12 r^5 \pi \Phi^2 (\varepsilon + 1) (\Phi_r r - 2 \Phi) v_{r,r}^2 + 8 r \left(\right. \\
& - \frac{\Phi_r^3 r^3}{8} + r^2 \left(r \left(\frac{3 \Phi}{8} + (\varepsilon + 1) r^2 \pi \right) v_r + \frac{3 \Phi}{4} + (4 \varepsilon + 4) \pi r^2 \right) \Phi_r^2 \\
& + r \left(r^4 \pi (\varepsilon + 1) v_r^2 - 4 r \left(\frac{3 \Phi}{8} + (\varepsilon + 1) r^2 \pi \right) v_r - \frac{3 \Phi}{2} + (-14 \varepsilon \right. \\
& - 14) \pi r^2 \left. \right) \Phi \Phi_r - 2 \left(r^4 \pi (\varepsilon + 1) v_r^2 - \frac{5 r \left(\frac{3 \Phi}{10} + (\varepsilon + 1) r^2 \pi \right) v_r}{2} - \frac{\Phi}{2} \right.
\end{aligned}$$

$$\begin{aligned}
& + (-5\varepsilon - 5)\pi r^2 \left. \right) \Phi^2 \left. \right) \Phi v_{r,r} + (r^4 v_r - 2r^3) \Phi_r^4 + 4r^2 \left(r^2 \left(-\frac{\Phi}{4} + (\varepsilon \right. \right. \\
& + 1) r^2 \pi \left. \right) v_r^2 - \Phi v_r r + \frac{5\Phi}{2} + (-4\varepsilon - 4)\pi r^2 \left. \right) \Phi_r^3 - 4r \left(r^3 \left(-\frac{\Phi}{4} + (\varepsilon \right. \right. \\
& + 1) r^2 \pi \left. \right) v_r^3 - \frac{3\Phi v_r^2 r^2}{2} - 8r \left(\frac{\Phi}{16} + (\varepsilon + 1) r^2 \pi \right) v_r + 4\Phi + (-8\varepsilon \\
& - 8)\pi r^2 \left. \right) \Phi \Phi_r^2 + 4 \left(r^6 \pi (\varepsilon + 1) v_r^4 + 2r^3 \left(-\frac{\Phi}{2} + (\varepsilon + 1) r^2 \pi \right) v_r^3 \right. \\
& - 14r^2 \left(\frac{3\Phi}{14} + (\varepsilon + 1) r^2 \pi \right) v_r^2 - 12r \left(-\frac{\Phi}{6} + (\varepsilon + 1) r^2 \pi \right) v_r + 2\Phi + (\\
& - 4\varepsilon - 4)\pi r^2 \left. \right) \Phi^2 \Phi_r - 8v_r \left(r^5 \pi (\varepsilon + 1) v_r^3 - r^2 \left(\frac{\Phi}{2} + (\varepsilon + 1) r^2 \pi \right) v_r^2 \right. \\
& - 6r \left(\frac{\Phi}{6} + (\varepsilon + 1) r^2 \pi \right) v_r + \Phi + (-2\varepsilon - 2)\pi r^2 \left. \right) \Phi^3 \left. \right) / \left(4r^2 \left(-2r \left(\frac{\Phi}{8} \right. \right. \right. \\
& + (\varepsilon + 1) r^2 \pi \left. \right) \Phi_r + \left(r^3 \pi (\varepsilon + 1) v_r + \frac{\Phi}{2} + (2\varepsilon + 2)\pi r^2 \right) \Phi \left. \right) (\Phi_r r \\
& - 2\Phi) \Phi \left. \right) \& \text{where} \left[-\Phi_r r + 2\Phi \neq 0, r^2 \Phi (v_r r - 2) \Phi_{r,r} - (\Phi_r r \right. \\
& \left. - 2\Phi) (\Phi v_{r,r} r^2 + (-r^2 v_r + 2r) \Phi_r + \Phi (r^2 v_r^2 - 2)) \neq 0 \right]
\end{aligned}$$

> Case[2]

$$\left[\lambda_r = \left(-24576 r \left(-\frac{9 r^2 (\varepsilon + 1) \Phi \pi \left(\frac{\Phi}{12} + (\varepsilon + 1) r^2 \pi \right)^3 e^\lambda}{16} + \left(-\frac{\Phi^4}{1536} \right. \right. \right. \right. \tag{3.4}$$

$$\left. \left. \left. - \frac{r^2 \pi (\varepsilon + 1) \Phi^3}{64} - \frac{3 r^4 \pi^2 (\varepsilon + 1)^2 \Phi^2}{32} + \frac{r^6 \pi^3 (\varepsilon + 1)^3 \Phi}{6} + r^8 \pi^4 (\varepsilon \right. \right. \right.$$

$$\begin{aligned}
& + 1)^4 \left(\frac{\Phi}{16} + (\varepsilon + 1) r^2 \pi \right) \Phi_r - 36864 \Phi \left(\left(\frac{\Phi^2}{16} + \frac{7 r^2 \pi (\varepsilon + 1) \Phi}{8} \right. \right. \\
& \left. \left. + r^4 \pi^2 (\varepsilon + 1)^2 \right) r^2 (\varepsilon + 1) \pi \left(\frac{\Phi}{12} + (\varepsilon + 1) r^2 \pi \right)^2 e^\lambda + \frac{\Phi^5}{18432} \right. \\
& \left. + \frac{19 r^2 \pi (\varepsilon + 1) \Phi^4}{9216} + \frac{59 r^4 \pi^2 (\varepsilon + 1)^2 \Phi^3}{2304} + \frac{25 r^6 \pi^3 (\varepsilon + 1)^3 \Phi^2}{288} \right. \\
& \left. - \frac{5 r^8 \pi^4 (\varepsilon + 1)^4 \Phi}{18} - \frac{2 r^{10} \pi^5 (\varepsilon + 1)^5}{3} \right) \Bigg/ \left(2304 r^4 (\varepsilon + 1) \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi \Phi_r \pi \left(\frac{\Phi}{12} + (\varepsilon + 1) r^2 \pi \right)^2 \right), v_r \\
& = \frac{8 r \left(\frac{\Phi}{8} + (\varepsilon + 1) r^2 \pi \right) \Phi_r - 8 \left(\frac{\Phi}{4} + (\varepsilon + 1) r^2 \pi \right) \Phi}{4 \pi \Phi r^3 (\varepsilon + 1)}, \Phi_r^2 \\
& = \frac{16 \left(\frac{\Phi^2}{16} + \frac{7 r^2 \pi (\varepsilon + 1) \Phi}{8} + r^4 \pi^2 (\varepsilon + 1)^2 \right) (\Phi_r r - \Phi)}{3 r^2 \left(\frac{\Phi}{12} + (\varepsilon + 1) r^2 \pi \right)} \Bigg] \& \text{where } \left[-\Phi_r r \right. \\
& \left. + 2 \Phi \neq 0 \right]
\end{aligned}$$

> Case[3]

$$\begin{aligned}
& \left[e^{-\lambda} = -\frac{4 \pi r^2 (\varepsilon + 1)}{\Phi}, \lambda_r = 0, v_{r,r} \right. \\
& \left. = \frac{-r^4 \pi (\varepsilon + 1) v_r^2 + 2 r^3 \pi (\varepsilon + 1) v_r + \Phi + (4 \varepsilon + 4) \pi r^2}{2 r^4 \pi (\varepsilon + 1)}, \Phi_r = \frac{2 \Phi}{r} \right] \& \text{where} \\
& \left[\Phi \neq 0 \right]
\end{aligned} \tag{3.5}$$

Although an exact solution for the first two cases is difficult to obtain, Case[3] can be tackled and

solved to the end in few steps

> $sys[3] := op(1, Case[3])$

$$sys_3 := \left[e^{-\lambda} = -\frac{4\pi r^2(\epsilon+1)}{\Phi}, \lambda_r = 0, v_{r,r} \right. \\ \left. = \frac{-r^4\pi(\epsilon+1)v_r^2 + 2r^3\pi(\epsilon+1)v_r + \Phi + (4\epsilon+4)\pi r^2}{2r^4\pi(\epsilon+1)}, \Phi_r = \frac{2\Phi}{r} \right] \quad (3.6)$$

The first equation is just a constraint between the functions $\lambda(r)$ and $\Phi(r)$

> $constraint, subsystem := selectremove(has, sys[3], exp)$

$$constraint, subsystem := \left[e^{-\lambda} = -\frac{4\pi r^2(\epsilon+1)}{\Phi} \right], \left[\lambda_r = 0, v_{r,r} \right. \\ \left. = \frac{-r^4\pi(\epsilon+1)v_r^2 + 2r^3\pi(\epsilon+1)v_r + \Phi + (4\epsilon+4)\pi r^2}{2r^4\pi(\epsilon+1)}, \Phi_r = \frac{2\Phi}{r} \right] \quad (3.7)$$

The subsystem of differential equations can now be solved in closed form in terms of three arbitrary constants $_C1, _C2, _C3$

> $sol_{subsystem} := dsolve(subsystem, explicit)$

$$sol_{subsystem} := \left\{ \Phi = _C1 r^2, v = \right. \\ \left. -\frac{1}{\sqrt{\pi(\epsilon+1)}} \left(\ln \left(\frac{(32\epsilon+32)\pi + 4_C1}{\left(\pi(\epsilon+1) \left(r \frac{\sqrt{(8\epsilon+8)\pi + _C1}}{\sqrt{\pi(\epsilon+1)}} \right) _C2 - _C3 \right)^2} \right) \sqrt{\pi(\epsilon+1)} \right. \right. \\ \left. \left. + \ln(r) \left(\sqrt{(8\epsilon+8)\pi + _C1} - 2\sqrt{\pi(\epsilon+1)} \right) \right), \lambda = _C2 \right\} \quad (3.8)$$

Introducing this value of Phi(r) into the constraint we specialize one of these constants, we choose $_C1$

> $eval(constraint, sol_{subsystem})$

$$\left[e^{-_C2} = -\frac{4\pi(\epsilon+1)}{_C1} \right] \quad (3.9)$$

> $PDEtools:-Solve((3.9), _C1)$

$$\{ _C1 = -4\pi(\epsilon+1)e^{-_C2} \} \quad (3.10)$$

The exact solution to the system constructed at the beginning,

> ode_{sys}

$$\left[\frac{1}{8r^2} \left((2\Phi_{r,r}r^2 - 2\Phi v_{r,r}r^2 + (r^2 v_r - r^2 \lambda_r - 4r)\Phi_r - \Phi(r^2 v_r^2 - r^2 \lambda_r v_r \right. \right. \\ \left. \left. - 2r^3 \pi(\epsilon+1)v_r + \Phi + (4\epsilon+4)\pi r^2) \right) \right] \quad (3.11)$$

$$\begin{aligned}
& -4) e^{-\lambda} - 4\Phi + (-16\varepsilon - 16)\pi r^2), \frac{1}{8r^2} \left((-6\Phi_{r,r} r^2 + 2\Phi v_{r,r} r^2 \right. \\
& \left. + \Phi v_r^2 r^2 - r(\Phi \lambda_r r - \Phi_r r + 4\Phi) v_r + (3r^2 \Phi_r - 4r\Phi) \lambda_r + 4\Phi_r r - 4\Phi \right) \\
& e^{-\lambda} + 4\Phi + (-16\varepsilon - 16)\pi r^2), \frac{1}{8r^2} \left((2\Phi_{r,r} r^2 + 2\Phi v_{r,r} r^2 + \Phi v_r^2 r^2 \right. \\
& \left. - r(\Phi \lambda_r r + 3\Phi_r r - 4\Phi) v_r + (-r^2 \Phi_r + 4r\Phi) \lambda_r + 4\Phi_r r - 4\Phi \right) e^{-\lambda} + 4\Phi \\
& \left. + (48\varepsilon + 48)\pi r^2 \right\}
\end{aligned}$$

is then

> $sol[3] := subs(\mathbf{(3.10)}, sol_{subsystem})$

$$\begin{aligned}
sol_3 := & \left\{ \Phi = -4\pi(\varepsilon + 1)e^{-C_2} r^2, v = \right. & \mathbf{(3.12)} \\
& -\frac{1}{\sqrt{\pi(\varepsilon + 1)}} \left(\ln \left(\frac{(32\varepsilon + 32)\pi - 16\pi(\varepsilon + 1)e^{-C_2}}{\left(\pi(\varepsilon + 1) \left(r \frac{\sqrt{(8\varepsilon + 8)\pi - 4\pi(\varepsilon + 1)e^{-C_2}}}{\sqrt{\pi(\varepsilon + 1)}} \right)^2} \right)} \right) \right. \\
& \left. \sqrt{\pi(\varepsilon + 1)} + \ln(r) \left(\sqrt{(8\varepsilon + 8)\pi - 4\pi(\varepsilon + 1)e^{-C_2}} - 2\sqrt{\pi(\varepsilon + 1)} \right) \right), \lambda \\
& = \left. \begin{matrix} _C2 \end{matrix} \right\}
\end{aligned}$$

Verifying this result

> $odetest(sol[3], ode_{sys})$

{0}

(3.13)